
Correct by Construction

Dijkstra: “The only effective way to raise the
confidence level of a program significantly is
to give a convincing proof of its correctness.
But one should not first make the program and
then prove its correctness, because then the re-
quirement of providing the proof would only
increase the poor programmer’s burden. On
the contrary: the programmer should . . . ”

“. . . [let] correctness proof and program grow
hand in hand: with the choice of the structure
of the correctness proof one designs a program
for which this proof is applicable.”

Deriving Programs from Specifications

• In functional program derivation, the specification
itself is a function, albeit probably not an efficient
one.

• From the specification we construct a function that
equals the specification.

• The calculation is the proof.

• In the previous class to proceed by expanding and
reducing the definitions, until we obtain an induc-
tive definition of the specification.

• But that does not work all the time.

• In this lecture we review some techniques that
might work for more cases.

1 Tupling

Steep Lists

• A steep list is a list in which every element is larger
than the sum of those to its right:

steep :: List Int → Bool
steep [ ] = True
steep (x : xs) = steep xs ∧ x > sum xs .

• The definition above, if executed directly, is an
O(n2) program. Can we do better?

• Just nowwe learned to construct a generalised func-
tion which takes more input. This time, we try the
dual technique: to construct a function returning
more results.

Generalise by Returning More

• Recall that fst (a, b) = a and snd (a, b) = b.

• It is hard to quickly compute steep alone. But if we
define

steepsum xs = (steep xs, sum xs),

• and manage to synthesise a quick definition of
steepsum , we can implement steep by steep = fst ·
steepsum .

• We again proceed by case analysis. Trivially,

steepsum [ ] = (True, 0).

Deriving for the Non-Empty Case

For the case for non-empty inputs:

steepsum (x : xs)
= { definition of steepsum }

(steep (x : xs), sum (x : xs))
= { definitions of steep and sum }

(steep xs ∧ x > sum xs, x+ sum xs)
= { extracting sub-expressions involving xs }

let (b, y) = (steep xs, sum xs)
in (b ∧ x > y, x+ y)

= { definition of steepsum }
let (b, y) = steepsum xs
in (b ∧ x > y, x+ y).

Synthesised Program

• We have thus come up with a O(n) time program:

steep = fst · steepsum
steepsum [ ] = (True, 0)
steepsum (x : xs) = let (b, y) = steepsum xs

in (b ∧ x > y, x+ y),

• Again we observe the phenomena that a more gen-
eral function is easier to implement.

2 Accumulating Parameters
Reversing a List

• The function reverse is defined by:

reverse [ ] = [ ],
reverse (x : xs) = reverse xs ++[x].

• E.g. reverse [1, 2, 3, 4] =
((([ ] ++[4])++[3])++[2])++[1] = [4, 3, 2, 1].

• But how about its time complexity? Since (++) is
O(n), it takes O(n2) time to revert a list this way.

• Can we make it faster?

2.1 Fast List Reversal
Introducing an Accumulating Parameter

• Let us consider a generalisation of reverse . Define:

revcat :: [a]→ [a]→ [a]
revcat xs ys = reverse xs ++ ys .

• If we can construct a fast implementation of revcat ,
we can implement reverse by:

reverse xs = revcat xs [ ].

Reversing a List, Base Case
Let us use our old trick. Consider the case when xs is

[ ]:

revcat [ ] ys
= { definition of revcat }

reverse [ ] ++ ys
= { definition of reverse }

[ ] ++ ys
= { definition of (++) }

ys.

Reversing a List, Inductive Case
Case x : xs :

revcat (x : xs) ys
= { definition of revcat }

reverse (x : xs)++ ys
= { definition of reverse }

(reverse xs ++[x]) ++ ys
= { since (xs ++ ys)++ zs = xs ++(ys ++ zs) }

reverse xs ++([x] ++ ys)
= { definition of revcat }

revcat xs (x : ys).

Linear-Time List Reversal

• We have therefore constructed an implementation
of revcat which runs in linear time!

revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys).

• A generalisation of reverse is easier to implement
than reverse itself? How come?

• If you try to understand revcat operationally, it is
not difficult to see how it works.

– The partially reverted list is accumulated in ys.

– The initial value of ys is set by reverse xs =
revcat xs [ ].

– Hmm... it is like a loop, isn’t it?

2.2 Tail Recursion and Loops

Tracing Reverse

reverse [1, 2, 3, 4]
= revcat [1, 2, 3, 4] [ ]
= revcat [2, 3, 4] [1]
= revcat [3, 4] [2, 1]
= revcat [4] [3, 2, 1]
= revcat [ ] [4, 3, 2, 1]
= [4, 3, 2, 1]

reverse xs = revcat xs [ ]
revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys)

xs, ys ← XS , [ ];
while xs ̸= [ ] do

xs, ys ← (tail xs), (head xs : ys);
return ys

Tail Recursion

• Tail recursion: a special case of recursion in which
the last operation is the recursive call.

f x1 . . . xn = {base case}
f x1 . . . xn = f x′

1 . . . x′
n

• To implement general recursion, we need to keep a
stack of return addresses. For tail recursion, we do
not need such a stack.

• Tail recursive definitions are like loops. Each xi is
updated to x′

i in the next iteration of the loop.

• The first call to f sets up the initial values of each
xi.

Accumulating Parameters

• To efficiently perform a computation (e.g.
reverse xs), we introduce a generalisation with an
extra parameter, e.g.:

revcat xs ys = reverse xs ++ ys .

• Try to derive an efficient implementation of the gen-
eralised function. The extra parameter is usually
used to “accumulate" some results, hence the name.

– To make the accumulation work, we usually
need some kind of associativity.

• A technique useful for, but not limited to, construct-
ing tail-recursive definition of functions.

Accumulating Parameter: Another Example

• Recall the “sum of squares" problem:

sumsq [ ] = 0
sumsq (x : xs) = square x+ sumsq xs .

• The program still takes linear space (for the stack of
return addresses). Let us construct a tail recursive
auxiliary function.

• Introduce ssp xs n = sumsq xs + n.

• Initialisation: sumsq xs = ssp xs 0.

• Construct ssp:

ssp [ ] n = 0 + n = n
ssp (x : xs) n = (square x+ sumsq xs) + n

= sumsq xs + (square x+ n)
= ssp xs (square x+ n).

2.3 BeingQuicker by Doing More!
BeingQuicker by Doing More?

• A more generalised program can be implemented
more efficiently?

– A common phenomena! Sometimes the less
general function cannot be implemented in-
ductively at all!

– It also often happens that a theorem needs to
be generalised to be proved. We will see that
later.

• An obvious question: how do we know what gener-
alisation to pick?

• There is no easy answer — finding the right gener-
alisation one of the most difficulty act in program-
ming!

• For the past few examples, we choose the generali-
sation to exploit associativity.

• Sometimes we simply generalise by examining the
form of the formula.

Labelling a List

• Consider the task of labelling elements in a list with
its index.

index :: List a→ List (Int , a)
index = zip [0..]

• To construct an inductive definition, the case for [ ]
is easy. For the x : xs case:

index (x : xs)
= zip [0..] (x : xs)
= (0, x) : zip [1..] xs

• Alas, zip [1..] cannot be fold back to index !

• What if we turn the varying part into. . . a variable?

Labelling a List, Second Attempt

• Introduce idxFrom :: List a → Int →
List (Int , a):

idxFrom xs n = zip [n..] xs

• Initialisation: index xs = idxFrom xs 0.

• We reason:

idxFrom (x : xs) n
= zip [n..] (x : xs)
= (n, x) : zip [n+ 1..] xs
= (n, x) : idxFrom xs (n+ 1)

3 Proof by Strengthening
Summing Up a List in Reverse

• Prove: sum · reverse = sum , using the defini-
tion reverse xs = revcat xs [ ]. That is, proving
sum (revcat xs [ ]) = sum xs.

• Base case trivial. For the case x : xs :

sum (reverse (x : xs))
= sum (revcat (x : xs) [ ])
= sum (revcat xs [x])

• Then we are stuck, since we cannot use the induc-
tion hypothesis sum (revcat xs [ ]) = sum xs.

• Again, generalise [x] to a variable.

Summing Up a List in Reverse, Second Attempt

• Second attempt: prove a lemma:

sum (revcat xs ys) = sum xs + sum ys

• By letting ys = [ ] we get the previous property.

• For the case x : xs we reason:

sum (revcat (x : xs) ys)
= sum (revcat xs (x : ys))
= { induction hypothesis }

sum xs + sum (x : ys)
= sum xs + x+ sum ys
= sum (x : xs) + sum ys

Work Less by Proving More

• A stronger theorem is easier to prove! Why is that?

• By strengthening the theorem, we also have a
stronger induction hypothesis, which makes an in-
ductive proof possible.

– Finding the right generalisation is an art — it’s
got to be strong enough to help the proof, yet
not too strong to be provable.

• The same with programming. By generalising a
function with additional arguments, it is passed
more information it may use, thus making an in-
ductive definition possible.

– The speeding up of revcat , in retrospect, is an
accidental “side effect” — revcat , being induc-
tive, goes through the list only once, and is
therefore quicker.

A Real Case

• A property I actually had to prove for a paper:

smsp (trim (x : xs)) = smsp (trim (x : win xs))
⇐ smsp (trim (x : xs)) >d mds xs

• It took me a week to construct the right generalisa-
tion:

smsp (trim (zs ++ xs)) = smsp (trim (zs ++win xs))
⇐ smsp (trim (zs ++ xs)) >d mds xs

• Once the right property is found, the actual proof
was done in about 20 minutes.

• “Someone once described research as ‘finding out
something to find out, then finding it out’; the first
part is often harder than the second.”

Remark

• The sum ·reverse example is superficial — the same
property is much easier to prove using the O(n2)-
time definition of reverse .

• That’s one of the reason we defer the discussion
about efficiency — to prove properties of a function
we sometimes prefer to roll back to a slower version.

• In our exercises there is an example where you need
revcat to prove a property about reverse .

– Show that reverse · reverse = id

